
A1 Either a property or its negation is positive, but not both: ∀φ[P (¬φ)↔ ¬P (φ)]
A2 A property necessarily implied

by a positive property is positive: ∀φ∀ψ[(P (φ) ∧�∀x[φ(x)→ ψ(x)])→ P (ψ)]

T1 Positive properties are possibly exemplified: ∀φ[P (φ)→ ♦∃xφ(x)]
D1 A God-like being possesses all positive properties: G(x)↔ ∀φ[P (φ)→ φ(x)]

A3 The property of being God-like is positive: P (G)

C Possibly, God exists: ♦∃xG(x)
A4 Positive properties are necessarily positive: ∀φ[P (φ)→ � P (φ)]

D2 An essence of an individual is a property possessed by it
and necessarily implying any of its properties:

φ ess. x↔ φ(x) ∧ ∀ψ(ψ(x)→ �∀y(φ(y)→ ψ(y)))

T2 Being God-like is an essence of any God-like being: ∀x[G(x)→ G ess. x]

D3 Necessary existence of an individual is
the necessary exemplification of all its essences: NE (x)↔ ∀φ[φ ess. x→ �∃yφ(y)]

A5 Necessary existence is a positive property: P (NE )

T3 Necessarily, God exists: �∃xG(x)

Figure 1: Scott’s version of Gödel’s ontological argument [12].

1 Introduction

Dana Scott’s version [12] (cf. Fig. 1) of Gödel’s proof of God’s existence [8] is formalized
in quantified modal logic KB (QML KB) within the proof assistant Isabelle/HOL. QML
KB is modeled as a fragment of classical higher-order logic (HOL); thus, the formalization is
essentially a formalization in HOL. The employed embedding of QML KB in HOL is adapting
the work of Benzmüller and Paulson [2, 1]. Note that the QML KB formalization employs
quantification over individuals and quantification over sets of individuals (properties).
The gaps in Scott’s proof have been automated with Sledgehammer [5], performing remote
calls to the higher-order automated theorem prover LEO-II [3]. Sledgehammer suggests the
Metis [9] calls, which result in proofs that are verified by Isabelle/HOL. For consistency check-
ing, the model finder Nitpick [6] has been employed. The successfull calls to Sledgehammer
are deliberately kept as comments in the file for demonstration purposes (normally, they are
automatically eliminated by Isabelle/HOL).

Isabelle is described in the textbook by Nipkow, Paulson, and Wenzel [10] and in tutorials
available at: http://isabelle.in.tum.de.

1.1 Related Work

The formalization presented here is related to the THF [14] and Coq [4] formalizations at
https://github.com/FormalTheology/GoedelGod/tree/master/Formalizations/.
An older ontological argument by Anselm was formalized in PVS by John Rushby [11].

2 An Embedding of QML KB in HOL

The types i for possible worlds and µ for individuals are introduced.
typedecl i — the type for possible worlds
typedecl µ — the type for indiviuals

Possible worlds are connected by an accessibility relation r.
consts r :: i ⇒ i ⇒ bool (infixr ‹r› 70 ) — accessibility relation r

QML formulas are translated as HOL terms of type i ⇒ bool. This type is abbreviated as σ.
type-synonym σ = (i ⇒ bool)

The classical connectives ¬,∧,→, and ∀ (over individuals and over sets of individuals) and
∃ (over individuals) are lifted to type σ. The lifted connectives are m¬, m∧, m→, ∀ , and
∃ (the latter two are modeled as constant symbols). Other connectives can be introduced
analogously. We exemplarily do this for m∨ , m≡, and mL= (Leibniz equality on individuals).
Moreover, the modal operators � and ♦ are introduced. Definitions could be used instead of
abbreviations.

abbreviation mnot :: σ ⇒ σ (‹m¬›) where m¬ ϕ ≡ (λw. ¬ ϕ w)
abbreviation mand :: σ ⇒ σ ⇒ σ (infixr ‹m∧› 65 ) where ϕ m∧ ψ ≡ (λw. ϕ w ∧ ψ w)
abbreviation mor :: σ ⇒ σ ⇒ σ (infixr ‹m∨› 70 ) where ϕ m∨ ψ ≡ (λw. ϕ w ∨ ψ w)
abbreviation mimplies :: σ ⇒ σ ⇒ σ (infixr ‹m→› 74 ) where ϕ m→ ψ ≡ (λw. ϕ w −→ ψ w)
abbreviation mequiv:: σ ⇒ σ ⇒ σ (infixr ‹m≡› 76 ) where ϕ m≡ ψ ≡ (λw. ϕ w ←→ ψ w)
abbreviation mforall :: ( ′a ⇒ σ) ⇒ σ (‹∀ ›) where ∀ Φ ≡ (λw. ∀ x. Φ x w)
abbreviation mexists :: ( ′a ⇒ σ) ⇒ σ (‹∃ ›) where ∃ Φ ≡ (λw. ∃ x. Φ x w)
abbreviation mLeibeq :: µ ⇒ µ ⇒ σ (infixr ‹mL=› 90 ) where x mL= y ≡ ∀ (λϕ. (ϕ x m→ ϕ y))
abbreviation mbox :: σ ⇒ σ (‹�›) where � ϕ ≡ (λw. ∀ v. w r v −→ ϕ v)
abbreviation mdia :: σ ⇒ σ (‹♦›) where ♦ ϕ ≡ (λw. ∃ v. w r v ∧ ϕ v)

For grounding lifted formulas, the meta-predicate valid is introduced.
abbreviation valid :: σ ⇒ bool (‹[-]›) where [p] ≡ ∀w. p w

3 Gödel’s Ontological Argument

Constant symbol P (Gödel’s ‘Positive’) is declared.
consts P :: (µ ⇒ σ) ⇒ σ

The meaning of P is restricted by axioms A1 (a/b): ∀φ[P (¬φ) ↔ ¬P (φ)] (Either a property
or its negation is positive, but not both.) and A2 : ∀φ∀ψ[(P (φ)∧�∀x[φ(x)→ ψ(x)])→ P (ψ)]
(A property necessarily implied by a positive property is positive).

axiomatization where
A1a: [∀ (λΦ. P (λx. m¬ (Φ x)) m→ m¬ (P Φ))] and
A1b: [∀ (λΦ. m¬ (P Φ) m→ P (λx. m¬ (Φ x)))] and
A2 : [∀ (λΦ. ∀ (λΨ. (P Φ m∧ � (∀ (λx. Φ x m→ Ψ x))) m→ P Ψ))]

We prove theorem T1: ∀φ[P (φ)→ ♦∃xφ(x)] (Positive properties are possibly exemplified). T1
is proved directly by Sledgehammer with command sledgehammer [provers = remote-leo2 ].
Sledgehammer suggests to call Metis with axioms A1a and A2. Metis sucesfully generates a
proof object that is verified in Isabelle/HOL’s kernel.

theorem T1 : [∀ (λΦ. P Φ m→ ♦ (∃ Φ))]
— sledgehammer [provers = remote_leo2]
〈proof 〉

Next, the symbol G for ‘God-like’ is introduced and defined as G(x)↔ ∀φ[P (φ)→ φ(x)]
(A God-like being possesses all positive properties).

definition G :: µ ⇒ σ where G = (λx. ∀ (λΦ. P Φ m→ Φ x))

Axiom A3 is added: P (G) (The property of being God-like is positive). Sledgehammer and
Metis then prove corollary C : ♦∃xG(x) (Possibly, God exists).

axiomatization where A3 : [P G]

corollary C : [♦ (∃ G)]
— sledgehammer [provers = remote_leo2]
〈proof 〉

Axiom A4 is added: ∀φ[P (φ)→ � P (φ)] (Positive properties are necessarily positive).
axiomatization where A4 : [∀ (λΦ. P Φ m→ � (P Φ))]

Symbol ess for ‘Essence’ is introduced and defined as

φ ess. x↔ φ(x) ∧ ∀ψ(ψ(x)→ �∀y(φ(y)→ ψ(y)))

(An essence of an individual is a property possessed by it and necessarily implying any of its
properties).

definition ess :: (µ ⇒ σ) ⇒ µ ⇒ σ (infixr ‹ess› 85 ) where
Φ ess x = Φ x m∧ ∀ (λΨ. Ψ x m→ � (∀ (λy. Φ y m→ Ψ y)))

Next, Sledgehammer and Metis prove theorem T2 : ∀x[G(x)→ G ess. x]
(Being God-like is an essence of any God-like being).

theorem T2 : [∀ (λx. G x m→ G ess x)]
— sledgehammer [provers = remote_leo2]
〈proof 〉

Symbol NE, for ‘Necessary Existence’, is introduced and defined as

NE (x)↔ ∀φ[φ ess. x→ �∃yφ(y)]

(Necessary existence of an individual is the necessary exemplification of all its essences).
definition NE :: µ ⇒ σ where NE = (λx. ∀ (λΦ. Φ ess x m→ � (∃ Φ)))

Moreover, axiom A5 is added: P (NE ) (Necessary existence is a positive property).
axiomatization where A5 : [P NE ]

The B axiom (symmetry) for relation r is stated. B is needed only for proving theorem T3
and for corollary C2.

axiomatization where sym: x r y −→ y r x

Finally, Sledgehammer and Metis prove the main theorem T3 : �∃xG(x)
(Necessarily, God exists).

theorem T3 : [� (∃ G)]
— sledgehammer [provers = remote_leo2]
〈proof 〉

Surprisingly, the following corollary can be derived even without the T axiom (reflexivity).
corollary C2 : [∃ G]
— sledgehammer [provers = remote_leo2]
〈proof 〉

The consistency of the entire theory is confirmed by Nitpick.
lemma True nitpick [satisfy, user-axioms, expect = genuine] 〈proof 〉

4 Additional Results on Gödel’s God.

Gödel’s God is flawless: (s)he does not have non-positive properties.
theorem Flawlessness: [∀ (λΦ. ∀ (λx. (G x m→ (m¬ (P Φ) m→ m¬ (Φ x)))))]
— sledgehammer [provers = remote_leo2]
〈proof 〉

There is only one God: any two God-like beings are equal.
theorem Monotheism: [∀ (λx.∀ (λy. (G x m→ (G y m→ (x mL= y)))))]
— sledgehammer [provers = remote_leo2]
〈proof 〉

5 Modal Collapse

Gödel’s axioms have been criticized for entailing the so-called modal collapse. The prover
Satallax [7] confirms this. However, sledgehammer is not able to determine which axioms,
definitions and previous theorems are used by Satallax; hence it suggests to call Metis using
everything, but this (unsurprinsingly) fails. Attempting to use ‘Sledegehammer min’ to min-
imize Sledgehammer’s suggestion does not work. Calling Metis with T2, T3 and ess-def also
does not work.

lemma MC : [∀ (λΦ.(Φ m→ (� Φ)))]
— sledgehammer [provers = remote_satallax]
— by (metis T2 T3 ess_def)
〈proof 〉

